Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

External Utilities

```
In[∘]:= HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Internal Utilities

Canonical Form:

```
In[∘]:= CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@Together[Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CF[x]];
```

The Kronecker δ:

```
In[∘]:= Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

```
In[∘]:= 𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]_$k_ := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Zip and Bind

Variables and their duals:

```
In[∘]:= {t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)_i;
```

Finite Zips: (* Perhaps switch Expand to Collect[__, ζ]? *)

```
In[∘]:= expand[sd_SeriesData] := MapAt[expand, sd, 3];
expand[ℰ_] := Expand[ℰ];
Zip_{}[P_] := P;
Zip_{ζ_,ζs___}[P_] := (expand[P // Zip_{ζs}] /. f_. ζ^d_. :→ ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

In[∘]:=
```
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
  zs = Table[ζ*, {ζ, ζs}];
  c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
  ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
  ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
  qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}Q, {ζ, ζs}, {z, zs}];
  zrule = Thread[zs → qt.(zs + ys)];
  Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
  simp /@ 𝔼[L, Q2, Det[qt] e^{-Q2} Zip_{ζs}[e^{Q1} (P /. zrule)]] ];
QZip_{ζs_List} := QZip_{ζs,CF};
```

Upper to lower and lower to Upper:

In[∘]:=
```
U2l = {B_{i_}^{p_.} → e^{-p ħ γ b_i}, B^{p_.} → e^{-p ħ γ b}, T_{i_}^{p_.} → e^{p ħ t_i}, T^{p_.} → e^{p ħ t}, 𝒜_{i_}^{p_.} → e^{p γ α_i}, 𝒜^{p_.} → e^{p γ α}};
l2U = {e^{c_. b_i_+d_.} :→ B_i^{-c/(ħ γ)} e^d, e^{c_. b+d_.} :→ B^{-c/(ħ γ)} e^d,
  e^{c_. t_i_+d_.} :→ T_i^{c/ħ} e^d, e^{c_. t+d_.} :→ T^{c/ħ} e^d,
  e^{c_. α_i_+d_.} :→ 𝒜_i^{c/γ} e^d, e^{c_. α+d_.} :→ 𝒜^{c/γ} e^d,
  e^{ℰ_} :→ e^{Expand@ℰ}};
```

LZip implements the “L-level zips” on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^{Q} as a single”P”. Here the z’s are b and α and the ζ’s are β and a.

In[∘]:=
```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
  zs = Table[ζ*, {ζ, ζs}];
  c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
  ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
  ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
  lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ}L, {ζ, ζs}, {z, zs}];
  zrule = Thread[zs → lt.(zs + ys)];
  L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
  Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
  simp /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_{ζs}[e^{L1+Q1} (P /. U2l /. zrule)]] //. l2U ];
LZip_{ζs_List} := LZip_{ζs,CF};
```

In[∘]:=
```
Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := Module[{n},
  Times[
    L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
    R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
  ] // LZip_{Flatten@Table[{β_{n@i}, τ_{n@i}, a_{n@i}}, {i, {is}}]} // QZip_{Flatten@Table[{ξ_{n@i}, y_{n@i}}, {i, {is}}]} ];
B_{L_List}[L_, R_] := Bind_L[L, R]; B_{is___}[L_, R_] := Bind_{is}[L, R];
```

Booting Up

In[∘]:=
```
$k = 2;
```

In[∘]:=

```
Boot[kk_] := (If[kk > 0, Boot[kk - 1]];
   Block[{$k = kk, i, j, k, m, n, tu = TimeUsed[]},
    am_{i_,j_->k_,$k} = 𝔼[(α_i + α_j) a_k, (e^{-γ α_j} ξ_i + ξ_j) x_k, 1]_{$k};
    bm_{i_,j_→k_,$k} = 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, e^{(e^{-ϵ β_i}-1) η_j y_k}]_{$k};
    R_{i_,j_,$k} = 𝔼[ħ a_j b_i, ħ x_j y_i, e^(Sum_{k=2}^{$k+1} ((1 - e^{γ ϵ ħ})^k (ħ y_i x_j)^k)/(k (1 - e^{k γ ϵ ħ})))]_{$k};
    P_{i_,j_,$k} = If[$k == 0, 𝔼[β_i α_j / ħ, η_i ξ_j / ħ, 1]_0,
      MapAt[(# - ϵ^{$k} Coefficient[(R_{n,m} ~ B_{n,m} ~ ((P_{n,j,0})_{$k} (P_{i,m,$k-1})_{$k}))[[3]], ϵ, $k]) &,
       (P_{i,j,$k-1})_{$k}, 3]];
    aS_{i_,$k} = 𝔼[-α_i a_j, -ξ_i x_i,
       e^{ξ_i x_i} Sum[Expand[(-ħ γ ϵ)^k/(2^k k!) Nest[Expand[x_i^2 ∂_{{x_i,2}} #] &, e^{-ξ_i e^{ħ ϵ a_i} x_i}, k]], {k, 0, $k}]]_{$k};
      B_{i,j} ~ am_{i,j→i};
    aS̄_{i_,$k} = If[$k == 0, 𝔼[-a_i α_i, -x_i 𝒜_i ξ_i, 1]_0,
      MapAt[(# - ϵ^{$k} Coefficient[((aS̄_{i,0})_{$k} ~ B_i ~ aS_i ~ B_i ~ (aS̄_{i,$k-1})_{$k})[[3]], ϵ, $k]) &,
       (aS̄_{i,$k-1})_{$k}, 3]];
    (* t==ϵa-γb and b==-t/γ+ϵa/γ: *)
    b2t_{i_,$k} = 𝔼[α_i a_i - β_i t_i / γ, ξ_i x_i + η_i y_i, e^{ϵ β_i a_i/γ}]_{$k};
    t2b_{i_,$k} = 𝔼[α_i a_j - τ_i γ b_j, ξ_i x_j + η_i y_j, e^{ϵ τ_i a_j}]_{$k};
    Boot[$k] = Print["Booted @ $k=", $k, " in ", TimeUsed[] - tu, " sec."];
   ]);
```

In[∘]:=

```
am_{i_,j_→k_} := am_{i,j→k,$k}; am_{i_,j_→k_,$k_} := (Boot[$k]; am_{i,j→k,$k});
bm_{i_,j_→k_} := bm_{i,j→k,$k}; bm_{i_,j_→k_,$k_} := (Boot[$k]; bm_{i,j→k,$k});
R_{i_,j_} := R_{i,j,$k}; R_{i_,j_,$k_} := (Boot[$k]; R_{i,j,$k});
P_{i_,j_} := P_{i,j,$k}; P_{i_,j_,$k_} := (Boot[$k]; P_{i,j,$k});
aS_{i_} := aS_{i,$k}; aS_{i_,$k_} := (Boot[$k]; aS_{i,$k});
aS̄_{i_} := aS̄_{i,$k}; aS̄_{i_,$k_} := (Boot[$k]; aS̄_{i,$k});
b2t_{i_} := b2t_{i,$k}; b2t_{i_,$k_} := (Boot[$k]; b2t_{i,$k});
t2b_{i_} := t2b_{i,$k}; t2b_{i_,$k_} := (Boot[$k]; t2b_{i,$k});
```

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, R → R_{i,j}, P → P_{i,j}, aS → aS_i, \overline{aS} → \overline{aS}_i, b2t → b2t_i, t2b → t2b_i
  }] // Column
```

Out[∘]=

$$\text{am} \to \mathbb{E}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\,\left(e^{-\gamma\,\alpha_j}\,\xi_i+\xi_j\right),\ 1\right]$$

$$\text{bm} \to \mathbb{E}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$R \to \mathbb{E}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$P \to \mathbb{E}\left[\frac{\alpha_j\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_j}{\hbar},\ 1+\frac{\gamma\,\epsilon\,\eta_i^2\,\xi_j^2}{4\,\hbar}\right]$$

$$\text{aS} \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(-2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\text{aS}} \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1-\frac{1}{2}\,\epsilon\left(-2\,\gamma\,\hbar\,x_i\,\mathcal{A}_i\,\xi_i+2\,\hbar\,a_i\,x_i\,\mathcal{A}_i\,\xi_i+\gamma\,\hbar\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\right]$$

$$\text{b2t} \to \mathbb{E}\left[a_i\,\alpha_i-\frac{t_i\,\beta_i}{\gamma},\ y_i\,\eta_i+x_i\,\xi_i,\ 1+\frac{a_i\,\beta_i\,\epsilon}{\gamma}+O[\epsilon]^2\right]$$

$$\text{t2b} \to \mathbb{E}\left[a_j\,\alpha_i-\gamma\,b_j\,\tau_i,\ y_j\,\eta_i+x_j\,\xi_i,\ 1+a_j\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Associativity of am and bm:

In[∘]:=
```
Timing@Block[{$k = 3},
  HL /@ {(am_{1,2→1} ~ B_1 ~ am_{1,3→1}) ≡ (am_{2,3→2} ~ B_2 ~ am_{1,2→1}), (bm_{1,2→1} ~ B_1 ~ bm_{1,3→1}) ≡ (bm_{2,3→2} ~ B_2 ~ bm_{1,2→1})}
 ]
```

```
Booted @ $k=3 in 0.125 sec.
```

Out[∘]= {0.390625, {True, True}}

R and *P* are inverses:

In[∘]:=
```
Timing@Block[{$k = 3}, {R_{i,j}, P_{i,k}, HL[R_{i,j} ~ B_i ~ P_{i,k} ≡ 𝔼[a_j α_k, x_j ξ_k, 1]]}]
```

Out[∘]=

$$\left\{0.03125,\ \left\{\mathbb{E}\left[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ 1-\frac{1}{4}\left(\gamma\,\hbar^3\,x_j^2\,y_i^2\right)\epsilon+\frac{1}{288}\left(32\,\gamma^2\,\hbar^5\,x_j^3\,y_i^3+9\,\gamma^2\,\hbar^6\,x_j^4\,y_i^4\right)\epsilon^2+\frac{1}{1152}\left(24\,\gamma^3\,\hbar^5\,x_j^2\,y_i^2-72\,\gamma^3\,\hbar^7\,x_j^4\,y_i^4-32\,\gamma^3\,\hbar^8\,x_j^5\,y_i^5-3\,\gamma^3\,\hbar^9\,x_j^6\,y_i^6\right)\epsilon^3+O[\epsilon]^4\right],\right.\right.$$

$$\left.\left.\mathbb{E}\left[\frac{\alpha_k\,\beta_i}{\hbar},\ \frac{\eta_i\,\xi_k}{\hbar},\ 1+\frac{\gamma\,\eta_i^2\,\xi_k^2\,\epsilon}{4\,\hbar}+\frac{\left(36\,\gamma^2\,\hbar^2\,\eta_i^2\,\xi_k^2+40\,\gamma^2\,\hbar\,\eta_i^3\,\xi_k^3+9\,\gamma^2\,\eta_i^4\,\xi_k^4\right)\epsilon^2}{288\,\hbar^2}+\frac{1}{1152\,\hbar^3}\left(48\,\gamma^3\,\hbar^4\,\eta_i^2\,\xi_k^2+192\,\gamma^3\,\hbar^3\,\eta_i^3\,\xi_k^3+156\,\gamma^3\,\hbar^2\,\eta_i^4\,\xi_k^4+40\,\gamma^3\,\hbar\,\eta_i^5\,\xi_k^5+3\,\gamma^3\,\eta_i^6\,\xi_k^6\right)\epsilon^3+O[\epsilon]^4\right],\ \text{True}\right\}\right\}$$

as and $\overline{\text{aS}}$ are inverses:

In[∘]:=
```
Timing@HL[\overline{aS}_1 ~ B_1 ~ aS_1 ≡ 𝔼[a_1 α_1, x_1 ξ_1, 1]]
```

Out[∘]= {0.109375, True}

S

In[∘]:=
```
bS_i_ := R_{i,n} ~ B_n ~ aS_n ~ B_n ~ P_{i,n};
\overline{bS}_i_ := R_{i,n} ~ B_n ~ \overline{aS}_n ~ B_n ~ P_{i,n};
```

In[∘]:= `Timing@HL[` $\overline{bS}_1 \sim B_1 \sim bS_1 \equiv \mathbb{E}[b_1\beta_1, y_1\eta_1, 1]$ `]`

Out[∘]= {0.984375, True}

Δ

In[∘]:= `Block[{i, j, k, l, m, n},` $a\Delta_{i\_\to j\_,k\_} = (R_{n,j} R_{m,k}) \sim B_{n,m} \sim bm_{n,m\to l} \sim B_l \sim P_{l,i}$`];`
$a\Delta_{i\to j,k}$

Booted @ $k=0 in 0.047 sec.

Booted @ $k=1 in 0.187 sec.

Booted @ $k=2 in 0.359 sec.

Out[∘]= $\mathbb{E}\Big[a_j\alpha_i + a_k\alpha_i,\ x_j\xi_i + x_k\xi_i,\ 1 + \frac{1}{2}\left(-2\hbar a_j x_k \xi_i + \gamma\hbar x_j x_k \xi_i^2\right)\epsilon + \frac{1}{24}\left(12\hbar^2 a_j^2 x_k \xi_i + 6\gamma^2\hbar^2 x_j x_k \xi_i^2 - 12\gamma\hbar^2 a_j x_j x_k \xi_i^2 + 12\hbar^2 a_j^2 x_k^2 \xi_i^2 + 4\gamma^2\hbar^2 x_j^2 x_k \xi_i^3 + 4\gamma^2\hbar^2 x_j x_k^2 \xi_i^3 - 12\gamma\hbar^2 a_j x_j x_k^2 \xi_i^3 + 3\gamma^2\hbar^2 x_j^2 x_k^2 \xi_i^4\right)\epsilon^2 + O[\epsilon]^3\Big]$

In[∘]:= `Block[{i, j, k, l, m, n},` $b\Delta_{i\_\to j\_,k\_} = (R_{j,n} R_{k,m}) \sim B_{n,m} \sim am_{n,m\to l} \sim B_l \sim P_{i,l}$`];`
$b\Delta_{i\to j,k}$

Out[∘]= $\mathbb{E}\Big[b_j\beta_i + b_k\beta_i,\ B_k y_j\eta_i + y_k\eta_i,\ 1 + \frac{1}{2}\gamma\hbar B_k y_j y_k \eta_i^2\epsilon + \frac{1}{24}\left(6\gamma^2\hbar^2 B_k y_j y_k \eta_i^2 + 4\gamma^2\hbar^2 B_k^2 y_j^2 y_k \eta_i^3 + 4\gamma^2\hbar^2 B_k y_j y_k^2 \eta_i^3 + 3\gamma^2\hbar^2 B_k^2 y_j^2 y_k^2 \eta_i^4\right)\epsilon^2 + O[\epsilon]^3\Big]$

The two halves

First check that on the generators this agrees with our conventions in the handout:

In[∘]:=
```
Timing@{{"[a,x]" → ((𝔼[0, 0, a₂ x₁] ~B₁,₂~ am₁,₂→₁)⟦3⟧ − (𝔼[0, 0, a₁ x₂] ~B₁,₂~ am₁,₂→₁)⟦3⟧),
    "[b,y]" → ((𝔼[0, 0, y₂ b₁] ~B₁,₂~ bm₁,₂→₁)⟦3⟧ − (𝔼[0, 0, y₁ b₂] ~B₁,₂~ bm₁,₂→₁)⟦3⟧)} /.
   z_1 → z,
  {"Δ[y]" → Last[𝔼[0, 0, y₁] ~B₁~ bΔ₁→₁,₂],
   "Δ[b]" → Last[𝔼[0, 0, b₁] ~B₁~ bΔ₁→₁,₂],
   "Δ[a]" → Last[𝔼[0, 0, a₁] ~B₁~ aΔ₁→₁,₂],
   "Δ[x]" → Last[𝔼[0, 0, x₁] ~B₁~ aΔ₁→₁,₂]},
  {
    "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ aS₁)⟦3⟧),
    "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ aS₁)⟦3⟧),
    "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ bS₁)⟦3⟧),
    "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ bS₁)⟦3⟧)
  } /. z_1 → z}
```

Out[∘]= $\{0.875, \{\{[a,x] \to -x\gamma, [b,y] \to -y\epsilon + O[\epsilon]^3\},$
$\{\Delta[y] \to (B_2 y_1 + y_2) + O[\epsilon]^3, \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3, \Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,$
$\Delta[x] \to (x_1 + x_2) - \hbar a_1 x_2 \epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2 \epsilon^2 + O[\epsilon]^3\}, \{S(a) \to -a + O[\epsilon]^3,$
$S(x) \to -x - a x \hbar \epsilon - \frac{1}{2}(a^2 x \hbar^2)\epsilon^2 + O[\epsilon]^3, S(b) \to -b + O[\epsilon]^3, S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately

(co)-associativity on both sides

In[∘]:=
```
Timing[HL /@
  {(aΔ₁→₁,₂ ~B₂~ aΔ₂→₂,₃) ≡ (aΔ₁→₁,₃ ~B₁~ aΔ₁→₁,₂), (bΔ₁→₁,₂ ~B₂~ bΔ₂→₂,₃) ≡ (bΔ₁→₁,₃ ~B₁~ bΔ₁→₁,₂),
   (am₁,₂→₁ ~B₁~ am₁,₃→₁) ≡ (am₂,₃→₂ ~B₂~ am₁,₂→₁), (bm₁,₂→₁ ~B₁~ bm₁,₃→₁) ≡ (bm₂,₃→₂ ~B₂~ bm₁,₂→₁)}]
```

Out[∘]= {0.40625, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:=
```
Timing[HL /@ {am₁,₂→₁ ~B₁~ aΔ₁→₁,₂ ≡ (aΔ₁→₁,₃ aΔ₂→₂,₄) ~B₁,₂,₃,₄~ (am₃,₄→₂ am₁,₂→₁),
    bm₁,₂→₁ ~B₁~ bΔ₁→₁,₂ ≡ (bΔ₁→₁,₃ bΔ₂→₂,₄) ~B₁,₂,₃,₄~ (bm₃,₄→₂ bm₁,₂→₁)}]
```

Out[∘]= {0.703125, {True, True}}

S is convolution inverse of id

In[∘]:=
```
Timing[HL[# ≡ 𝔼[0, 0, 1]] & /@ {
    (aΔ₁→₁,₂ ~B₁~ aS₁) ~B₁,₂~ am₁,₂→₁, (aΔ₁→₁,₂ ~B₂~ aS₂) ~B₁,₂~ am₁,₂→₁,
    (bΔ₁→₁,₂ ~B₁~ bS₁) ~B₁,₂~ bm₁,₂→₁, (bΔ₁→₁,₂ ~B₂~ bS₂) ~B₁,₂~ bm₁,₂→₁}]
```

Out[∘]= {1.26563, {True, True, True, True}}

S is an algebra anti-(co)morphism

In[∘]:=
```
Timing[HL /@ {am₁,₂→₁ ~B₁~ aS₁ ≡ (aS₁ aS₂) ~B₁,₂~ am₂,₁→₁, bm₁,₂→₁ ~B₁~ bS₁ ≡ (bS₁ bS₂) ~B₁,₂~ bm₂,₁→₁,
    aS₁ ~B₁~ aΔ₁→₁,₂ ≡ aΔ₁→₂,₁ ~B₁,₂~ (aS₁ aS₂), bS₁ ~B₁~ bΔ₁→₁,₂ ≡ bΔ₁→₂,₁ ~B₁,₂~ (bS₁ bS₂)}]
```

Out[∘]= {2.84375, {True, True, True, True}}

Pairing axioms

In[]:=
```
Timing[HL /@ {(bm_{1,2→1} 𝔼[α3 a3, ξ3 x3, 1]) ~B_{1,3}~P_{1,3} ≡
   (𝔼[β1 b1, η1 y1, 1] 𝔼[β2 b2, η2 y2, 1] aΔ_{3→4,5}) ~B_{1,4}~P_{1,4}~B_{2,5}~P_{2,5},
  (bΔ_{1→1,2} 𝔼[α3 a3, ξ3 x3, 1] 𝔼[α4 a4, ξ4 x4, 1]) ~B_{1,3}~P_{1,3}~B_{2,4}~P_{2,4} ≡
   (𝔼[β1 b1, η1 y1, 1] am_{3,4→3}) ~B_{1,3}~P_{1,3} }]
```

Out[]= {0.984375, {True, True}}

In[]:=
```
Timing[HL /@ {(bS1 𝔼[α2 a2, ξ2 x2, 1]) ~B_{1,2}~P_{1,2} ≡ (𝔼[β1 b1, η1 y1, 1] aS2) ~B_{1,2}~P_{1,2},
  (bS̄1 𝔼[α2 a2, ξ2 x2, 1]) ~B_{1,2}~P_{1,2} ≡ (𝔼[β1 b1, η1 y1, 1] aS̄2) ~B_{1,2}~P_{1,2}}]
```

Out[]= {1.89063, {True, True}}

The Double

The double multiplication (should really bind the a’s and b’s separately)

In[•]:=

```
Block[{i, j, k}, dm_{i_,j_→k_} =
  (𝔼[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1] (aΔ_{i→h1,h2} ~ B_{h2} ~ aΔ_{h2→h2,h3}) (bΔ_{j→t1,t2} ~ B_{t2} ~ bΔ_{t2→t2,t3})) ~
   B_{h3} ~ \overline{aS}_{h3} ~ B_{t1,h3} ~ (P_{t1,h3}) ~ B_{t3,h1} ~ (P_{t3,h1}) ~ B_{h2,j,i,t2} ~ (am_{h2,j→k} bm_{i,t2→k})]
```

Out[•]=

$$\mathbb{E}\Big[a_k\,\alpha_i + a_k\,\alpha_j + b_k\,\beta_i + b_k\,\beta_j,\ \frac{1}{\hbar\,\mathcal{A}_i\,\mathcal{A}_j}$$
$$(\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_i + \hbar\,y_k\,\mathcal{A}_j\,\eta_j + \hbar\,x_k\,\mathcal{A}_i\,\xi_i + \mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i - B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i + \hbar\,x_k\,\mathcal{A}_i\,\mathcal{A}_j\,\xi_j),$$
$$1 + \frac{1}{4\,\hbar\,\mathcal{A}_i\,\mathcal{A}_j}\Big(-4\,\hbar\,y_k\,\mathcal{A}_j\,\beta_i\,\eta_j - 4\,\hbar\,x_k\,\mathcal{A}_i\,\beta_j\,\xi_i + 4\,\gamma\,\hbar^2\,x_k\,y_k\,\eta_j\,\xi_i +$$
$$4\,\hbar\,a_k\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i + 2\,\gamma\,\hbar\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i - 6\,\gamma\,\hbar\,B_k\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i + 2\,\gamma\,\hbar\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2 -$$
$$6\,\gamma\,\hbar\,B_k\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2 + \gamma\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 - 4\,\gamma\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 + 3\,\gamma\,B_k^2\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2\Big)\,\epsilon + \frac{1}{288\,\hbar^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2}$$
$$\Big(144\,\hbar^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i^2\,\eta_j + 144\,\hbar^2\,y_k^2\,\mathcal{A}_j^2\,\beta_i^2\,\eta_j^2 + 144\,\hbar^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j^2\,\xi_i + 144\,\gamma^2\,\hbar^4\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i -$$
$$144\,\hbar^3\,a_k^2\,B_k\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j\,\xi_i - 288\,\gamma\,\hbar^3\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_i\,\eta_j\,\xi_i - 288\,\gamma\,\hbar^3\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_j\,\eta_j\,\xi_i +$$
$$288\,\hbar^2\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_i\,\beta_j\,\eta_j\,\xi_i + 144\,\gamma^2\,\hbar^4\,x_k\,y_k^2\,\mathcal{A}_j\,\eta_j^2\,\xi_i + 72\,\gamma^2\,\hbar^3\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i -$$
$$360\,\gamma^2\,\hbar^3\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i + 432\,\gamma\,\hbar^3\,a_k\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i - 288\,\gamma\,\hbar^3\,x_k\,y_k^2\,\mathcal{A}_j\,\beta_i\,\eta_j^2\,\xi_i -$$
$$144\,\gamma\,\hbar^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i\,\eta_j^2\,\xi_i + 432\,\gamma\,\hbar^2\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i\,\eta_j^2\,\xi_i - 288\,\hbar^2\,a_k\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i\,\eta_j^2\,\xi_i +$$
$$48\,\gamma^2\,\hbar^3\,y_k^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i - 336\,\gamma^2\,\hbar^3\,B_k\,y_k^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i - 144\,\gamma\,\hbar^2\,y_k^2\,\mathcal{A}_j^2\,\beta_i\,\eta_j^3\,\xi_i + 432\,\gamma\,\hbar^2\,B_k\,y_k^2\,\mathcal{A}_j^2\,\beta_i\,\eta_j^3\,\xi_i +$$
$$144\,\hbar^2\,x_k^2\,\mathcal{A}_i^2\,\beta_j^2\,\xi_i^2 + 144\,\gamma^2\,\hbar^4\,x_k^2\,y_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2 + 72\,\gamma^2\,\hbar^3\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j\,\xi_i^2 - 360\,\gamma^2\,\hbar^3\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j\,\xi_i^2 +$$
$$432\,\gamma\,\hbar^3\,a_k\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j\,\xi_i^2 - 288\,\gamma\,\hbar^3\,x_k^2\,y_k\,\mathcal{A}_i\,\beta_j\,\eta_j\,\xi_i^2 - 144\,\gamma\,\hbar^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j\,\eta_j\,\xi_i^2 +$$
$$432\,\gamma\,\hbar^2\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j\,\eta_j\,\xi_i^2 - 288\,\hbar^2\,a_k\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j\,\eta_j\,\xi_i^2 + 144\,\gamma^2\,\hbar^4\,x_k^2\,y_k^2\,\eta_j^2\,\xi_i^2 +$$
$$360\,\gamma^2\,\hbar^3\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 - 1512\,\gamma^2\,\hbar^3\,B_k\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 + 288\,\gamma\,\hbar^3\,a_k\,B_k\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2 +$$
$$36\,\gamma^2\,\hbar^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i^2 - 216\,\gamma^2\,\hbar^2\,B_k\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i^2 + 288\,\gamma\,\hbar^2\,a_k\,B_k\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i^2 +$$
$$180\,\gamma^2\,\hbar^2\,B_k^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i^2 - 432\,\gamma\,\hbar^2\,a_k\,B_k^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i^2 + 144\,\hbar^2\,a_k^2\,B_k^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^2\,\xi_i^2 -$$
$$144\,\gamma\,\hbar^2\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_i\,\eta_j^2\,\xi_i^2 + 432\,\gamma\,\hbar^2\,B_k\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_i\,\eta_j^2\,\xi_i^2 - 144\,\gamma\,\hbar^2\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_j\,\eta_j^2\,\xi_i^2 +$$
$$432\,\gamma\,\hbar^2\,B_k\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\beta_j\,\eta_j^2\,\xi_i^2 + 144\,\gamma^2\,\hbar^3\,x_k\,y_k^2\,\mathcal{A}_j\,\eta_j^3\,\xi_i^2 - 432\,\gamma^2\,\hbar^3\,B_k\,x_k\,y_k^2\,\mathcal{A}_j\,\eta_j^3\,\xi_i^2 +$$
$$120\,\gamma^2\,\hbar^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^2 - 816\,\gamma^2\,\hbar^2\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^2 + 144\,\gamma\,\hbar^2\,a_k\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^2 +$$
$$984\,\gamma^2\,\hbar^2\,B_k^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^2 - 432\,\gamma\,\hbar^2\,a_k\,B_k^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^2 - 72\,\gamma\,\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i\,\eta_j^3\,\xi_i^2 +$$
$$288\,\gamma\,\hbar\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i\,\eta_j^3\,\xi_i^2 - 216\,\gamma\,\hbar\,B_k^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\beta_i\,\eta_j^3\,\xi_i^2 + 36\,\gamma^2\,\hbar^2\,y_k^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^2 -$$
$$216\,\gamma^2\,\hbar^2\,B_k\,y_k^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^2 + 324\,\gamma^2\,\hbar^2\,B_k^2\,y_k^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^2 + 48\,\gamma^2\,\hbar^3\,x_k^2\,\mathcal{A}_i^2\,\eta_j\,\xi_i^3 - 336\,\gamma^2\,\hbar^3\,B_k\,x_k^2\,\mathcal{A}_i^2\,\eta_j\,\xi_i^3 -$$
$$144\,\gamma\,\hbar^2\,x_k^2\,\mathcal{A}_i^2\,\beta_j\,\eta_j\,\xi_i^3 + 432\,\gamma\,\hbar^2\,B_k\,x_k^2\,\mathcal{A}_i^2\,\beta_j\,\eta_j\,\xi_i^3 + 144\,\gamma^2\,\hbar^3\,x_k^2\,y_k\,\mathcal{A}_i\,\eta_j^2\,\xi_i^3 -$$
$$432\,\gamma^2\,\hbar^3\,B_k\,x_k^2\,y_k\,\mathcal{A}_i\,\eta_j^2\,\xi_i^3 + 120\,\gamma^2\,\hbar^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^2\,\xi_i^3 - 816\,\gamma^2\,\hbar^2\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^2\,\xi_i^3 +$$
$$144\,\gamma\,\hbar^2\,a_k\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^2\,\xi_i^3 + 984\,\gamma^2\,\hbar^2\,B_k^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^2\,\xi_i^3 - 432\,\gamma\,\hbar^2\,a_k\,B_k^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^2\,\xi_i^3 -$$
$$72\,\gamma\,\hbar\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j\,\eta_j^2\,\xi_i^3 + 288\,\gamma\,\hbar\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j\,\eta_j^2\,\xi_i^3 - 216\,\gamma\,\hbar\,B_k^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\beta_j\,\eta_j^2\,\xi_i^3 +$$
$$144\,\gamma^2\,\hbar^2\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^3\,\xi_i^3 - 720\,\gamma^2\,\hbar^2\,B_k\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^3\,\xi_i^3 + 864\,\gamma^2\,\hbar^2\,B_k^2\,x_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^3\,\xi_i^3 +$$
$$40\,\gamma^2\,\hbar\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 - 312\,\gamma^2\,\hbar\,B_k\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 + 72\,\gamma\,\hbar\,a_k\,B_k\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 + 600\,\gamma^2\,\hbar\,B_k^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 -$$
$$288\,\gamma\,\hbar\,a_k\,B_k^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 - 328\,\gamma^2\,\hbar\,B_k^3\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 + 216\,\gamma\,\hbar\,a_k\,B_k^3\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^3\,\xi_i^3 + 36\,\gamma^2\,\hbar\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^3 -$$
$$252\,\gamma^2\,\hbar\,B_k\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^3 + 540\,\gamma^2\,\hbar\,B_k^2\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^3 - 324\,\gamma^2\,\hbar\,B_k^3\,y_k\,\mathcal{A}_i\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^3 +$$
$$36\,\gamma^2\,\hbar^2\,x_k^2\,\mathcal{A}_i^2\,\eta_j^2\,\xi_i^4 - 216\,\gamma^2\,\hbar^2\,B_k\,x_k^2\,\mathcal{A}_i^2\,\eta_j^2\,\xi_i^4 + 324\,\gamma^2\,\hbar^2\,B_k^2\,x_k^2\,\mathcal{A}_i^2\,\eta_j^2\,\xi_i^4 + 36\,\gamma^2\,\hbar\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^3\,\xi_i^4 -$$
$$252\,\gamma^2\,\hbar\,B_k\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^3\,\xi_i^4 + 540\,\gamma^2\,\hbar\,B_k^2\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^3\,\xi_i^4 - 324\,\gamma^2\,\hbar\,B_k^3\,x_k\,\mathcal{A}_i^2\,\mathcal{A}_j\,\eta_j^3\,\xi_i^4 + 9\,\gamma^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^4 -$$
$$72\,\gamma^2\,B_k\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^4 + 198\,\gamma^2\,B_k^2\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^4 - 216\,\gamma^2\,B_k^3\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^4 + 81\,\gamma^2\,B_k^4\,\mathcal{A}_i^2\,\mathcal{A}_j^2\,\eta_j^4\,\xi_i^4\Big)\,\epsilon^2 + O[\epsilon]^3\Big]$$

In[•]:=

```
Block[{i}, dS_{i_} = 𝔼[β_i b_1 + α_i a_2, η_i y_1 + ξ_i x_2, 1] ~ B_{1,2} ~ (\overline{bS}_1 aS_2) ~ B_{1,2} ~ dm_{2,1→i}]
```

Out[•]=
$$\mathbb{E}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ \frac{-\hbar\,y_i\,\mathcal{A}_i\,\eta_i-\hbar\,B_i\,x_i\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\eta_i\,\xi_i-B_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{\hbar\,B_i},$$

$$1+\frac{1}{4\,\hbar\,B_i^2}\Big(4\,\gamma\,\hbar^2\,B_i\,y_i\,\mathcal{A}_i\,\eta_i-4\,\hbar\,B_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-2\,\gamma\,\hbar^2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-4\,\hbar^2\,a_i\,B_i^2\,x_i\,\mathcal{A}_i\,\xi_i-4\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-$$
$$4\,\gamma\,\hbar\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\hbar\,a_i\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,\gamma\,\hbar\,B_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i-4\,\gamma\,\hbar^2\,B_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i+$$
$$4\,B_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-4\,B_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+6\,\gamma\,\hbar\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar\,B_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,\gamma\,\hbar^2\,B_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+$$
$$6\,\gamma\,\hbar\,B_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-2\,\gamma\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-3\,\gamma\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+4\,\gamma\,B_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-\gamma\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\Big)\,\epsilon+$$

$$\frac{1}{288\,\hbar^2\,B_i^4}\Big(-144\,\gamma^2\,\hbar^4\,B_i^3\,y_i\,\mathcal{A}_i\,\eta_i+288\,\gamma\,\hbar^3\,B_i^3\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-144\,\hbar^2\,B_i^3\,y_i\,\mathcal{A}_i\,\beta_i^2\,\eta_i+$$
$$504\,\gamma^2\,\hbar^4\,B_i^2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-576\,\gamma\,\hbar^3\,B_i^2\,y_i^2\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2+144\,\hbar^2\,B_i^2\,y_i^2\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i^2-288\,\gamma^2\,\hbar^4\,B_i\,y_i^3\,\mathcal{A}_i^3\,\eta_i^3+$$
$$144\,\gamma\,\hbar^3\,B_i\,y_i^3\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3+36\,\gamma^2\,\hbar^4\,y_i^4\,\mathcal{A}_i^4\,\eta_i^4-144\,\hbar^4\,a_i^2\,B_i^4\,x_i\,\mathcal{A}_i\,\xi_i-288\,\hbar^3\,a_i\,B_i^4\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-$$
$$144\,\hbar^2\,B_i^4\,x_i\,\mathcal{A}_i\,\beta_i^2\,\xi_i+144\,\gamma^2\,\hbar^3\,B_i^3\,\mathcal{A}_i\,\eta_i\,\xi_i-288\,\gamma\,\hbar^3\,a_i\,B_i^3\,\mathcal{A}_i\,\eta_i\,\xi_i+144\,\hbar^3\,a_i^2\,B_i^3\,\mathcal{A}_i\,\eta_i\,\xi_i-$$
$$144\,\gamma^2\,\hbar^3\,B_i^4\,\mathcal{A}_i\,\eta_i\,\xi_i+432\,\gamma^2\,\hbar^4\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i-576\,\gamma\,\hbar^4\,a_i\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i-$$
$$288\,\gamma\,\hbar^2\,B_i^3\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+288\,\hbar^2\,a_i\,B_i^3\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+288\,\gamma\,\hbar^2\,B_i^4\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-$$
$$864\,\gamma\,\hbar^3\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i\,\xi_i+288\,\hbar^3\,a_i\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i\,\xi_i+144\,\hbar\,B_i^3\,\mathcal{A}_i\,\beta_i^2\,\eta_i\,\xi_i-$$
$$144\,\hbar\,B_i^4\,\mathcal{A}_i\,\beta_i^2\,\eta_i\,\xi_i+288\,\hbar^2\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i\,\xi_i-1512\,\gamma^2\,\hbar^3\,B_i^2\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i+$$
$$720\,\gamma\,\hbar^3\,a_i\,B_i^2\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i+648\,\gamma^2\,\hbar^3\,B_i^3\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-720\,\gamma^2\,\hbar^4\,B_i^2\,x_i\,y_i^2\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i+$$
$$144\,\gamma\,\hbar^4\,a_i\,B_i^2\,x_i\,y_i^2\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i+1440\,\gamma\,\hbar^2\,B_i^2\,y_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i-288\,\hbar^2\,a_i\,B_i^2\,y_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i-$$
$$864\,\gamma\,\hbar^2\,B_i^3\,y_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i+432\,\gamma\,\hbar^3\,B_i^2\,x_i\,y_i^2\,\mathcal{A}_i^3\,\beta_i\,\eta_i^2\,\xi_i-288\,\hbar\,B_i^2\,y_i\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i^2\,\xi_i+$$
$$288\,\hbar\,B_i^3\,y_i\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i^2\,\xi_i+1344\,\gamma^2\,\hbar^3\,B_i\,y_i^2\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i-144\,\gamma\,\hbar^3\,a_i\,B_i\,y_i^2\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i-$$
$$480\,\gamma^2\,\hbar^3\,B_i^2\,y_i^2\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i+144\,\gamma^2\,\hbar^4\,B_i\,x_i\,y_i^3\,\mathcal{A}_i^4\,\eta_i^3\,\xi_i-576\,\gamma\,\hbar^2\,B_i\,y_i^2\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i+$$
$$288\,\gamma\,\hbar^2\,B_i^2\,y_i^2\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i-216\,\gamma^2\,\hbar^3\,y_i^3\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i+72\,\gamma^2\,\hbar^3\,B_i\,y_i^3\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i+72\,\gamma^2\,\hbar^4\,B_i^4\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2-$$
$$288\,\gamma\,\hbar^4\,a_i\,B_i^4\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+144\,\hbar^4\,a_i^2\,B_i^4\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2-288\,\gamma\,\hbar^3\,B_i^4\,x_i^2\,\mathcal{A}_i^2\,\beta_i\,\xi_i^2+288\,\hbar^3\,a_i\,B_i^4\,x_i^2\,\mathcal{A}_i^2\,\beta_i\,\xi_i^2+$$
$$144\,\hbar^2\,B_i^4\,x_i^2\,\mathcal{A}_i^2\,\beta_i^2\,\xi_i^2-792\,\gamma^2\,\hbar^3\,B_i^3\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2+1152\,\gamma\,\hbar^3\,a_i\,B_i^3\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-288\,\hbar^3\,a_i^2\,B_i^3\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2+$$
$$216\,\gamma^2\,\hbar^3\,B_i^4\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-432\,\gamma\,\hbar^3\,a_i\,B_i^4\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-576\,\gamma^2\,\hbar^4\,B_i^3\,x_i^2\,y_i\,\mathcal{A}_i^3\,\eta_i\,\xi_i^2+$$
$$288\,\gamma\,\hbar^4\,a_i\,B_i^3\,x_i^2\,y_i\,\mathcal{A}_i^3\,\eta_i\,\xi_i^2+1152\,\gamma\,\hbar^2\,B_i^3\,x_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i\,\xi_i^2-576\,\hbar^2\,a_i\,B_i^3\,x_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i\,\xi_i^2-$$
$$576\,\gamma\,\hbar^2\,B_i^4\,x_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i\,\xi_i^2+288\,\hbar^2\,a_i\,B_i^4\,x_i\,\mathcal{A}_i^2\,\beta_i\,\eta_i\,\xi_i^2+432\,\gamma\,\hbar^3\,B_i^3\,x_i^2\,y_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i\,\xi_i^2-$$
$$288\,\hbar\,B_i^3\,x_i\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i\,\xi_i^2+288\,\hbar\,B_i^4\,x_i\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i\,\xi_i^2+756\,\gamma^2\,\hbar^2\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-720\,\gamma\,\hbar^2\,a_i\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+$$
$$144\,\hbar^2\,a_i^2\,B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-1080\,\gamma^2\,\hbar^2\,B_i^3\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+576\,\gamma\,\hbar^2\,a_i\,B_i^3\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+324\,\gamma^2\,\hbar^2\,B_i^4\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+$$
$$2232\,\gamma^2\,\hbar^3\,B_i^2\,x_i\,y_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^2-720\,\gamma\,\hbar^3\,a_i\,B_i^2\,x_i\,y_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^2-792\,\gamma^2\,\hbar^3\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^2+$$
$$144\,\gamma\,\hbar^3\,a_i\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^2+216\,\gamma^2\,\hbar^4\,B_i^2\,x_i^2\,y_i^2\,\mathcal{A}_i^4\,\eta_i^2\,\xi_i^2-720\,\gamma\,\hbar\,B_i^2\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i^2+$$
$$288\,\hbar\,a_i\,B_i^2\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i^2+1152\,\gamma\,\hbar\,B_i^3\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i^2-288\,\hbar\,a_i\,B_i^3\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i^2-432\,\gamma\,\hbar\,B_i^4\,\mathcal{A}_i^2\,\beta_i\,\eta_i^2\,\xi_i^2-$$
$$1152\,\gamma\,\hbar^2\,B_i^2\,x_i\,y_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^2\,\xi_i^2+576\,\gamma\,\hbar^2\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^2\,\xi_i^2+144\,B_i^2\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i^2\,\xi_i^2-$$
$$288\,B_i^3\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i^2\,\xi_i^2+144\,B_i^4\,\mathcal{A}_i^2\,\beta_i^2\,\eta_i^2\,\xi_i^2-1632\,\gamma^2\,\hbar^2\,B_i\,y_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^2+432\,\gamma\,\hbar^2\,a_i\,B_i\,y_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^2+$$
$$1680\,\gamma^2\,\hbar^2\,B_i^2\,y_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^2-144\,\gamma\,\hbar^2\,a_i\,B_i^2\,y_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^2-336\,\gamma^2\,\hbar^2\,B_i^3\,y_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^2-$$
$$648\,\gamma^2\,\hbar^3\,B_i\,x_i\,y_i^2\,\mathcal{A}_i^4\,\eta_i^3\,\xi_i^2+216\,\gamma^2\,\hbar^3\,B_i^2\,x_i\,y_i^2\,\mathcal{A}_i^4\,\eta_i^3\,\xi_i^2+648\,\gamma\,\hbar\,B_i\,y_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^2-$$
$$864\,\gamma\,\hbar\,B_i^2\,y_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^2+216\,\gamma\,\hbar\,B_i^3\,y_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^2+432\,\gamma^2\,\hbar^2\,y_i^2\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^2-360\,\gamma^2\,\hbar^2\,B_i\,y_i^2\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^2+$$
$$72\,\gamma^2\,\hbar^2\,B_i^2\,y_i^2\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^2-144\,\gamma^2\,\hbar^4\,B_i^4\,x_i^3\,\mathcal{A}_i^3\,\xi_i^3+144\,\gamma\,\hbar^4\,a_i\,B_i^4\,x_i^3\,\mathcal{A}_i^3\,\xi_i^3+144\,\gamma\,\hbar^3\,B_i^4\,x_i^3\,\mathcal{A}_i^3\,\beta_i\,\xi_i^3+$$
$$912\,\gamma^2\,\hbar^3\,B_i^3\,x_i^2\,\mathcal{A}_i^3\,\eta_i\,\xi_i^3-576\,\gamma\,\hbar^3\,a_i\,B_i^3\,x_i^2\,\mathcal{A}_i^3\,\eta_i\,\xi_i^3-336\,\gamma^2\,\hbar^3\,B_i^4\,x_i^2\,\mathcal{A}_i^3\,\eta_i\,\xi_i^3+$$
$$144\,\gamma\,\hbar^3\,a_i\,B_i^4\,x_i^2\,\mathcal{A}_i^3\,\eta_i\,\xi_i^3+144\,\gamma^2\,\hbar^4\,B_i^3\,x_i^3\,y_i\,\mathcal{A}_i^4\,\eta_i\,\xi_i^3-576\,\gamma\,\hbar^2\,B_i^3\,x_i^2\,\mathcal{A}_i^3\,\beta_i\,\eta_i\,\xi_i^3+$$
$$288\,\gamma\,\hbar^2\,B_i^4\,x_i^2\,\mathcal{A}_i^3\,\beta_i\,\eta_i\,\xi_i^3-1416\,\gamma^2\,\hbar^2\,B_i^2\,x_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^3+648\,\gamma\,\hbar^2\,a_i\,B_i^2\,x_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^3+$$
$$1392\,\gamma^2\,\hbar^2\,B_i^3\,x_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^3-432\,\gamma\,\hbar^2\,a_i\,B_i^3\,x_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^3-264\,\gamma^2\,\hbar^2\,B_i^4\,x_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^3+$$
$$72\,\gamma\,\hbar^2\,a_i\,B_i^4\,x_i\,\mathcal{A}_i^3\,\eta_i^2\,\xi_i^3-648\,\gamma^2\,\hbar^3\,B_i^2\,x_i^2\,y_i\,\mathcal{A}_i^4\,\eta_i^2\,\xi_i^3+216\,\gamma^2\,\hbar^3\,B_i^3\,x_i^2\,y_i\,\mathcal{A}_i^4\,\eta_i^2\,\xi_i^3+$$
$$648\,\gamma\,\hbar\,B_i^2\,x_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^2\,\xi_i^3-864\,\gamma\,\hbar\,B_i^3\,x_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^2\,\xi_i^3+216\,\gamma\,\hbar\,B_i^4\,x_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^2\,\xi_i^3+544\,\gamma^2\,\hbar\,B_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3-$$
$$216\,\gamma\,\hbar\,a_i\,B_i\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3-1104\,\gamma^2\,\hbar\,B_i^2\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3+288\,\gamma\,\hbar\,a_i\,B_i^2\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3+672\,\gamma^2\,\hbar\,B_i^3\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3-$$
$$72\,\gamma\,\hbar\,a_i\,B_i^3\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3-112\,\gamma^2\,\hbar\,B_i^4\,\mathcal{A}_i^3\,\eta_i^3\,\xi_i^3+864\,\gamma^2\,\hbar^2\,B_i\,x_i\,y_i\,\mathcal{A}_i^4\,\eta_i^3\,\xi_i^3-720\,\gamma^2\,\hbar^2\,B_i^2\,x_i\,y_i\,\mathcal{A}_i^4\,\eta_i^3\,\xi_i^3+$$
$$144\,\gamma^2\,\hbar^2\,B_i^3\,x_i\,y_i\,\mathcal{A}_i^4\,\eta_i^3\,\xi_i^3-216\,\gamma\,B_i\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^3+504\,\gamma\,B_i^2\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^3-360\,\gamma\,B_i^3\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^3+$$
$$72\,\gamma\,B_i^4\,\mathcal{A}_i^3\,\beta_i\,\eta_i^3\,\xi_i^3-324\,\gamma^2\,\hbar\,y_i\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^3+540\,\gamma^2\,\hbar\,B_i\,y_i\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^3-252\,\gamma^2\,\hbar\,B_i^2\,y_i\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^3+$$
$$36\,\gamma^2\,\hbar\,B_i^3\,y_i\,\mathcal{A}_i^4\,\eta_i^4\,\xi_i^3+36\,\gamma^2\,\hbar^4\,B_i^4\,x_i^4\,\mathcal{A}_i^4\,\xi_i^4-216\,\gamma^2\,\hbar^3\,B_i^3\,x_i^3\,\mathcal{A}_i^4\,\eta_i\,\xi_i^4+72\,\gamma^2\,\hbar^3\,B_i^4\,x_i^3\,\mathcal{A}_i^4\,\eta_i\,\xi_i^4+$$

$432\gamma^2\hbar^2 B_i^2 x_i^2 \mathcal{A}_i^4 \eta_i^2 \xi_i^4 - 360\gamma^2\hbar^2 B_i^3 x_i^2 \mathcal{A}_i^4 \eta_i^2 \xi_i^4 + 72\gamma^2\hbar^2 B_i^4 x_i^2 \mathcal{A}_i^4 \eta_i^2 \xi_i^4 - 324\gamma^2\hbar B_i x_i \mathcal{A}_i^4 \eta_i^3 \xi_i^4 + 540\gamma^2\hbar B_i^2 x_i \mathcal{A}_i^4 \eta_i^3 \xi_i^4 - 252\gamma^2\hbar B_i^3 x_i \mathcal{A}_i^4 \eta_i^3 \xi_i^4 + 36\gamma^2\hbar B_i^4 x_i \mathcal{A}_i^4 \eta_i^3 \xi_i^4 + 81\gamma^2 \mathcal{A}_i^4 \eta_i^4 \xi_i^4 - 216\gamma^2 B_i \mathcal{A}_i^4 \eta_i^4 \xi_i^4 + 198\gamma^2 B_i^2 \mathcal{A}_i^4 \eta_i^4 \xi_i^4 - 72\gamma^2 B_i^3 \mathcal{A}_i^4 \eta_i^4 \xi_i^4 + 9\gamma^2 B_i^4 \mathcal{A}_i^4 \eta_i^4 \xi_i^4) \epsilon^2 + O[\epsilon]^3]$

In[∘]:=
```
Block[{i, j, k}, dΔ_{i_→j_,k_} = (bΔ_{i→3,1} aΔ_{i→2,4}) ~B_{1,2,3,4}~ (dm_{3,4→k} dm_{1,2→j})]
```

Out[∘]=
$$\mathbb{E}\Big[a_j\alpha_i + a_k\alpha_i + b_j\beta_i + b_k\beta_i,\ y_j\eta_i + B_j y_k\eta_i + x_j\xi_i + x_k\xi_i,$$
$$1 + \frac{1}{2}\left(\gamma\hbar B_j y_j y_k \eta_i^2 - 2\hbar a_j x_k \xi_i + \gamma\hbar x_j x_k \xi_i^2\right)\epsilon +$$
$$\frac{1}{24}\left(6\gamma^2\hbar^2 B_j y_j y_k \eta_i^2 + 4\gamma^2\hbar^2 B_j y_j^2 y_k \eta_i^3 + 4\gamma^2\hbar^2 B_j^2 y_j y_k^2 \eta_i^3 + 3\gamma^2\hbar^2 B_j^2 y_j^2 y_k^2 \eta_i^4 + 12\hbar^2 a_j^2 x_k \xi_i - 12\gamma\hbar^2 a_j B_j x_k y_j y_k \eta_i^2 \xi_i + 6\gamma^2\hbar^2 x_j x_k \xi_i^2 - 12\gamma\hbar^2 a_j x_j x_k \xi_i^2 + 12\hbar^2 a_j^2 x_k^2 \xi_i^2 + 6\gamma^2\hbar^2 B_j x_j x_k y_j y_k \eta_i^2 \xi_i^2 + 4\gamma^2\hbar^2 x_j^2 x_k \xi_i^3 + 4\gamma^2\hbar^2 x_j x_k^2 \xi_i^3 - 12\gamma\hbar^2 a_j x_j x_k^2 \xi_i^3 + 3\gamma^2\hbar^2 x_j^2 x_k^2 \xi_i^4\right)\epsilon^2 + O[\epsilon]^3\Big]$$

First check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
Timing@{{
    "[a,y]" → ((𝔼[0, 0, y₂ a₁] ~B_{1,2}~ dm_{1,2→1})⟦3⟧ - (𝔼[0, 0, y₁ a₂] ~B_{1,2}~ dm_{1,2→1})⟦3⟧),
    "[b,x]"  → ((𝔼[0, 0, x₂ b₁] ~B_{1,2}~ dm_{1,2→1})⟦3⟧ - (𝔼[0, 0, x₁ b₂] ~B_{1,2}~ dm_{1,2→1})⟦3⟧),
    "xy-qyx" → ((𝔼[0, 0, x₁ y₂] ~B_{1,2}~ dm_{1,2→1})⟦3⟧ - (1 + ϵ) (𝔼[0, 0, y₁ x₂] ~B_{1,2}~ dm_{1,2→1})⟦3⟧)
   } /. {z_1 → z} // Expand // Factor,
  {
   "Δ(a)" → ((𝔼[0, 0, a₁] ~B₁~ dΔ_{1→1,2})⟦3⟧),
   "Δ(x)" → ((𝔼[0, 0, x₁] ~B₁~ dΔ_{1→1,2})⟦3⟧),
   "Δ(b)" → ((𝔼[0, 0, b₁] ~B₁~ dΔ_{1→1,2})⟦3⟧),
   "Δ(y)" → ((𝔼[0, 0, y₁] ~B₁~ dΔ_{1→1,2})⟦3⟧)
  } // Simplify,
  {
    "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ dS₁)⟦3⟧),
    "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ dS₁)⟦3⟧),
    "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ dS₁)⟦3⟧),
    "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ dS₁)⟦3⟧)
  } /. {z_1 → z} // Simplify
 }
```

Out[∘]=
$$\Big\{3.5, \Big\{\Big\{[a,y] \to -y\gamma + O[\epsilon]^3,\ [b,x] \to x\epsilon + O[\epsilon]^3,$$
$$xy\text{-}qyx \to \left(-x y + \frac{1 - B + x y\hbar}{\hbar}\right) + (a B - x y + x y\gamma\hbar)\epsilon + \frac{1}{2}\left(-a^2 B\hbar + x y\gamma^2\hbar^2\right)\epsilon^2 + O[\epsilon]^3\Big\},$$
$$\Big\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - \hbar a_1 x_2\epsilon + \frac{1}{2}\hbar^2 a_1^2 x_2\epsilon^2 + O[\epsilon]^3,$$
$$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^3\Big\},$$
$$\Big\{S(a) \to -a + O[\epsilon]^3,\ S(x) \to -x - a x\hbar\epsilon - \frac{1}{2}\left(a^2 x\hbar^2\right)\epsilon^2 + O[\epsilon]^3,$$
$$S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\gamma\hbar\epsilon}{B} - \frac{\left(y\gamma^2\hbar^2\right)\epsilon^2}{2B} + O[\epsilon]^3\Big\}\Big\}\Big\}$$

Hopf algebra axioms on double

(co)-associativity

In[∘]:= $\text{Timing}[\text{HL} /@ \{(d\Delta_{1\to1,2} \sim B_2 \sim d\Delta_{2\to2,3}) \equiv (d\Delta_{1\to1,3} \sim B_1 \sim d\Delta_{1\to1,2}),\ (dm_{1,2\to1} \sim B_1 \sim dm_{1,3\to1}) \equiv (dm_{2,3\to2} \sim B_2 \sim dm_{1,2\text{->}1})\}]$

Out[∘]= {7.67188, {True, True}}

Δ is an algebra morphism

In[∘]:= $\text{Timing@HL}[dm_{1,2\to1} \sim B_1 \sim d\Delta_{1\to1,2} \equiv (d\Delta_{1\to1,3}\ d\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (dm_{3,4\to2}\ dm_{1,2\to1})]$

Out[∘]= {14.9688, True}

S is convolution inverse of id

In[∘]:= $\text{Timing}[\text{HL}[\# \equiv \mathbb{E}[0, 0, 1]] \,\&\, /@ \{(d\Delta_{1\to1,2} \sim B_1 \sim dS_1) \sim B_{1,2} \sim dm_{1,2\to1},\ (d\Delta_{1\to1,2} \sim B_2 \sim dS_2) \sim B_{1,2} \sim dm_{1,2\to1}\}]$

Out[∘]= {12.4219, {True, True}}

S is a (co)-algebra anti-morphism

In[∘]:= $\text{Timing}[\text{HL} /@ \text{Expand} /@ \{dm_{1,2\to1} \sim B_1 \sim dS_1 \equiv (dS_1\ dS_2) \sim B_{1,2} \sim dm_{2,1\to1},\ dS_1 \sim B_1 \sim d\Delta_{1\to1,2} \equiv d\Delta_{1\to2,1} \sim B_{1,2} \sim (dS_1\ dS_2)\}]$

Out[∘]= {27.9844, {True, True}}

Quasi-triangular axiom 1:

In[∘]:= $\text{Timing@HL}[R_{1,2} \sim B_1 \sim d\Delta_{1\text{->}1,3} \equiv (R_{1,4}\ R_{3,2}) \sim B_{2,4} \sim dm_{2,4\to2}]$

Out[∘]= {0.765625, True}

Quasi-triangular axiom 2:

In[∘]:= $\text{Timing@HL}[((d\Delta_{1\text{->}1,2}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to1}\ dm_{2,4\to2})) \equiv ((d\Delta_{1\to2,1}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{3,1\to1}\ dm_{4,2\to2}))]$

Out[∘]= {12.8594, True}

Reidemeister 3:

In[∘]:= $\text{Timing@HL}[((R_{1,2}\ R_{4,3}\ R_{5,6}) \sim B_{1,4} \sim dm_{1,4\to1} \sim B_{2,5} \sim dm_{2,5\to2} \sim B_{3,6} \sim dm_{3,6\to3}) \equiv ((R_{1,6}\ R_{2,3}\ R_{4,5}) \sim B_{1,4} \sim dm_{1,4\to1} \sim B_{2,5} \sim dm_{2,5\to2} \sim B_{3,6} \sim dm_{3,6\to3})]$

Out[∘]= {6.4375, True}

In[∘]:= $\text{Block}[\{i, j\}, \overline{R}_{i\_,j\_} = \text{Expand} /@ R_{i,j} \sim B_j \sim dS_j]$

Out[∘]= $\mathbb{E}\Big[-\hbar\, a_j\, b_i,\ -\frac{\hbar\, x_j\, y_i}{B_i},\ 1 + \frac{(-4\,\hbar^2\, a_j\, B_i\, x_j\, y_i - 3\,\gamma\,\hbar^3\, x_j^2\, y_i^2)\,\epsilon}{4\, B_i^2} + \frac{1}{288\, B_i^4}(-144\,\hbar^3\, a_j^2\, B_i^3\, x_j\, y_i + 144\,\gamma^2\,\hbar^4\, B_i^2\, x_j^2\, y_i^2 - 432\,\gamma\,\hbar^4\, a_j\, B_i^2\, x_j^2\, y_i^2 + 144\,\hbar^4\, a_j^2\, B_i^2\, x_j^2\, y_i^2 - 320\,\gamma^2\,\hbar^5\, B_i\, x_j^3\, y_i^3 + 216\,\gamma\,\hbar^5\, a_j\, B_i\, x_j^3\, y_i^3 + 81\,\gamma^2\,\hbar^6\, x_j^4\, y_i^4)\,\epsilon^2 + O[\epsilon]^3\Big]$

Reidemeister 2

In[∘]:= $\text{Timing}[\text{HL} /@ \{(\overline{R}_{1,2}\ R_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to1}\ dm_{2,4\to2}),\ (R_{1,2}\ \overline{R}_{3,4}) \sim B_{1,2,3,4} \sim (dm_{1,3\to1}\ dm_{2,4\to2})\}]$

Out[∘]= $\{8.42188, \{\mathbb{E}[0, 0, 1 + O[\epsilon]^3],\ \mathbb{E}[0, 0, 1 + O[\epsilon]^3]\}\}$

Deriving the Drinfeld element u and its inverse $\overline{u}$

In[∘]:=
```
Block[{i}, {
  u_i_ = R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i},
  ū_i_ = R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→i}
 }]
```

Out[∘]=
$$\Big\{\mathbb{E}\Big[-\hbar a_i b_i,\ -\frac{\hbar x_i y_i}{B_i},\ B_i + \frac{1}{4 B_i}\left(-4\hbar a_i B_i^2 - 4\gamma\hbar^2 B_i x_i y_i - 4\hbar^2 a_i B_i x_i y_i - 3\gamma\hbar^3 x_i^2 y_i^2\right)\epsilon + \frac{1}{288 B_i^3}\left(144\hbar^2 a_i^2 B_i^4 - 144\gamma^2\hbar^3 B_i^3 x_i y_i + 144\hbar^3 a_i^2 B_i^3 x_i y_i - 144\gamma^2\hbar^4 B_i^2 x_i^2 y_i^2 + 72\gamma\hbar^4 a_i B_i^2 x_i^2 y_i^2 + 144\hbar^4 a_i^2 B_i^2 x_i^2 y_i^2 - 104\gamma^2\hbar^5 B_i x_i^3 y_i^3 + 216\gamma\hbar^5 a_i B_i x_i^3 y_i^3 + 81\gamma^2\hbar^6 x_i^4 y_i^4\right)\epsilon^2 + O[\epsilon]^3\Big],$$
$$\mathbb{E}\Big[\hbar a_i b_i,\ \hbar x_i y_i,\ \frac{1}{B_i} + \frac{\left(4\hbar a_i - 4\gamma\hbar^2 x_i y_i - \gamma\hbar^3 x_i^2 y_i^2\right)\epsilon}{4 B_i} + \frac{1}{288 B_i}\left(144\hbar^2 a_i^2 + 144\gamma^2\hbar^3 x_i y_i - 288\gamma\hbar^3 a_i x_i y_i + 288\gamma^2\hbar^4 x_i^2 y_i^2 - 72\gamma\hbar^4 a_i x_i^2 y_i^2 + 104\gamma^2\hbar^5 x_i^3 y_i^3 + 9\gamma^2\hbar^6 x_i^4 y_i^4\right)\epsilon^2 + O[\epsilon]^3\Big]\Big\}$$

u and $\overline{u}$ are inverses

In[∘]:=
```
Timing@HL[(u_1 ū_2) ~ B_{1,2} ~ dm_{1,2→1} ≡ 𝔼[0, 0, 1]]
```

Out[∘]= {1.53125, True}

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}.
It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

In[∘]:=
```
Block[{$k = 3}, ((u_1 ~ B_1 ~ dS_1) ū_2) ~ B_{1,2} ~ dm_{1,2→1}]
```

Out[∘]=
$$\mathbb{E}\left[0,\ 0,\ \frac{1}{B_1} + \frac{\hbar a_1 \epsilon}{B_1} + \frac{\hbar^2 a_1^2 \epsilon^2}{2 B_1} + O[\epsilon]^3\right]$$

(* Needs fixing! *) So in our case S(u) = u z so S(u)u = $u^2 z$ and v = u$z^{\frac{1}{2}}$ and finally C = uv^{-1} = $z^{-\frac{1}{2}} = e^{t_1/2}(1 - \epsilon a_1)$.

In[∘]:=
```
Block[{i},
 {CC_i_ = 𝔼[0, 0, B_i^{1/2} e^{-ϵ a_i/2} + O[ϵ]^2],
  C̄C̄_i_ = 𝔼[0, 0, B_i^{-1/2} e^{ϵ a_i/2} + O[ϵ]^2]
 }]
```

Out[∘]=
$$\left\{\mathbb{E}\left[0,\ 0,\ \sqrt{B_i} - \frac{1}{2}\left(a_i\sqrt{B_i}\right)\epsilon + O[\epsilon]^2\right],\ \mathbb{E}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}} + \frac{a_i\,\epsilon}{2\sqrt{B_i}} + O[\epsilon]^2\right]\right\}$$

In[∘]:=
```
Block[{i, j}, {
  Kink_i_ = (R_{1,3} \overline{CC}_2) ~B_{1,2}~dm_{1,2→1}~B_{1,3}~dm_{1,3→i},
  \overline{Kink}_j_ = (\overline{R}_{1,3} CC_2) ~B_{1,2}~dm_{1,2→1}~B_{1,3}~dm_{1,3→j}
 }]
```

Out[∘]=
$$\left\{\mathbb{E}\left[\hbar\, a_i\, b_i,\ \hbar\, x_i\, y_i,\ \frac{1}{\sqrt{B_i}}+\frac{\left(2\, a_i-\gamma\,\hbar^3\, x_i^2\, y_i^2\right)\epsilon}{4\sqrt{B_i}}+O[\epsilon]^2\right],\right.$$
$$\left.\mathbb{E}\left[-\hbar\, a_j\, b_j,\ -\frac{\hbar\, x_j\, y_j}{B_j},\ \sqrt{B_j}+\frac{\left(-2\, a_j\, B_j^2-4\,\hbar^2\, a_j\, B_j\, x_j\, y_j-3\,\gamma\,\hbar^3\, x_j^2\, y_j^2\right)\epsilon}{4\, B_j^{3/2}}+O[\epsilon]^2\right]\right\}$$

In[∘]:=
```
k2 = (R_{3,1} CC_2) ~B_{1,2}~dm_{1,2→1}~B_{1,3}~dm_{1,3→i} /. e → E;
k4 = (\overline{R}_{3,1} \overline{CC}_2) ~B_{1,2}~dm_{1,2→1}~B_{1,3}~dm_{1,3→j} /. e → E;
Simplify@{Kink_i ≡ k2, \overline{Kink}_j ≡ k4, (Kink_i \overline{Kink}_j) ~B_{i,j}~dm_{i,j→1}}
```

Out[∘]= $\{\text{True}, \text{True}, \mathbb{E}[0, 0, 1+O[\epsilon]^2]\}$

Reidemeister 2:

In[∘]:=
```
(R_{1,2} \overline{R}_{3,4}) ~B_{1,3}~dm_{1,3→1}~B_{2,4}~dm_{2,4→2}
```

Out[∘]= $\mathbb{E}[0, 0, 1+O[\epsilon]^2]$

Cyclic Reidemeister 2:

In[∘]:=
```
(R_{1,4} \overline{R}_{5,2} \overline{CC}_3) ~B_{2,4}~dm_{2,4→2}~B_{1,3}~dm_{1,3→1}~B_{1,5}~dm_{1,5→1} ≡ \overline{CC}_1
```

Out[∘]= True

Trefoil

In[∘]:=
```
Z = R_{1,5} R_{6,2} R_{3,7} \overline{CC}_4 \overline{Kink}_8 \overline{Kink}_9 \overline{Kink}_10;
Do[Z = Z~B_{1,r}~dm_{1,r→1}, {r, 2, 10}];
Simplify /@ Z
```

Out[∘]=
$$\mathbb{E}\left[0,\ 0,\ \frac{B_1}{1-B_1+B_1^2}+\left(B_1\left(-B_1+2\,B_1^2+2\,B_1^4+a_1\left(-1+B_1-B_1^3+B_1^4\right)-2\,x_1\,y_1-B_1^3\left(3+2\,x_1\,y_1\right)\right)\epsilon\right)\Big/\left(1-B_1+B_1^2\right)^3+O[\epsilon]^2\right]$$

```
Timing[
 Z = R_{1,5} R_{6,2} R_{3,7} \overline{CC}_4 \overline{Kink}_8 \overline{Kink}_9 \overline{Kink}_10;
 Do[Z = Z~B_{1,r}~dm_{1,r→1}, {r, 2, 10}];
 Simplify /@ Z ]
```

In[∘]:=
```
b2t_i_ := E[α_i a_i - β_i t_i, ξ_i x_i + η_i y_i, 1 + ϵ β_i a_i + O[ϵ]^2]
t2b_i_ := E[α_i a_i - τ_i b_i, ξ_i x_i + η_i y_i, 1 + ϵ τ_i a_i + O[ϵ]^2]
```

In[∘]:= $R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC}_4\, \overline{Kink}_8\, \overline{Kink}_9\, \overline{Kink}_{10}$

Out[∘]= $$\mathbb{E}\Big[a_5 b_1 + a_7 b_3 + a_2 b_6 - a_8 b_8 - a_9 b_9 - a_{10} b_{10},\; x_5 y_1 + x_7 y_3 + x_2 y_6 - \frac{x_8 y_8}{B_8} - \frac{x_9 y_9}{B_9} - \frac{x_{10} y_{10}}{B_{10}},\; \frac{\sqrt{B_8}\,\sqrt{B_9}\,\sqrt{B_{10}}}{\sqrt{B_4}} + \left(\sqrt{B_{10}}\left(\sqrt{B_9}\left(\sqrt{B_8}\left(\frac{a_4}{2\sqrt{B_4}} - \frac{x_5^2 y_1^2}{4\sqrt{B_4}} - \frac{x_7^2 y_3^2}{4\sqrt{B_4}} - \frac{x_2^2 y_6^2}{4\sqrt{B_4}}\right) + \frac{-2 a_8 B_8^2 - 4 a_8 B_8 x_8 y_8 - 3 x_8^2 y_8^2}{4\sqrt{B_4}\, B_8^{3/2}}\right) + \frac{\sqrt{B_8}\,\left(-2 a_9 B_9^2 - 4 a_9 B_9 x_9 y_9 - 3 x_9^2 y_9^2\right)}{4\sqrt{B_4}\, B_9^{3/2}}\right) + \frac{1}{4\sqrt{B_4}\, B_{10}^{3/2}}\sqrt{B_8}\,\sqrt{B_9}\,\left(-2 a_{10} B_{10}^2 - 4 a_{10} B_{10} x_{10} y_{10} - 3 x_{10}^2 y_{10}^2\right)\right) \epsilon + O[\epsilon]^2\Big]$$

In[∘]:= $\left(R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC}_4\, \overline{Kink}_8\, \overline{Kink}_9\, \overline{Kink}_{10}\right) \sim B_{Range[10]} \sim$ Product[$b2t_i$, {i, 10}]

Out[∘]= $$\mathbb{E}\Big[-a_5 t_1 - a_7 t_3 - a_2 t_6 + a_8 t_8 + a_9 t_9 + a_{10} t_{10},\; \frac{1}{T_8 T_9 T_{10}}\left(T_8 T_9 T_{10} x_5 y_1 + T_8 T_9 T_{10} x_7 y_3 + T_8 T_9 T_{10} x_2 y_6 - T_9 T_{10} x_8 y_8 - T_8 T_{10} x_9 y_9 - T_8 T_9 x_{10} y_{10}\right),\; \frac{\sqrt{T_8}\,\sqrt{T_9}\,\sqrt{T_{10}}}{\sqrt{T_4}} + \frac{1}{4\sqrt{T_4}\, T_8^{3/2} T_9^{3/2} T_{10}^{3/2}}\Big(4 a_4 T_8^2 T_9^2 T_{10}^2 + 4 a_1 a_5 T_8^2 T_9^2 T_{10}^2 + 4 a_2 a_6 T_8^2 T_9^2 T_{10}^2 + 4 a_3 a_7 T_8^2 T_9^2 T_{10}^2 - 4 a_8 T_8^2 T_9^2 T_{10}^2 - 4 a_8^2 T_8^2 T_9^2 T_{10}^2 - 4 a_9 T_8^2 T_9^2 T_{10}^2 - 4 a_9^2 T_8^2 T_9^2 T_{10}^2 - 4 a_{10} T_8^2 T_9^2 T_{10}^2 - 4 a_{10}^2 T_8^2 T_9^2 T_{10}^2 - T_8^2 T_9^2 T_{10}^2 x_5^2 y_1^2 - T_8^2 T_9^2 T_{10}^2 x_7^2 y_3^2 - T_8^2 T_9^2 T_{10}^2 x_2^2 y_6^2 - 8 a_8 T_8 T_9^2 T_{10}^2 x_8 y_8 - 3 T_9^2 T_{10}^2 x_8^2 y_8^2 - 8 a_9 T_8^2 T_9 T_{10}^2 x_9 y_9 - 3 T_8^2 T_{10}^2 x_9^2 y_9^2 - 8 a_{10} T_8^2 T_9^2 T_{10} x_{10} y_{10} - 3 T_8^2 T_9^2 x_{10}^2 y_{10}^2\Big) \epsilon + O[\epsilon]^2\Big]$$

In[∘]:= Z = $\left(\left(\left(R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC}_4\, \overline{Kink}_8\, \overline{Kink}_9\, \overline{Kink}_{10}\right) \sim B_{Range[10]} \sim \text{Product}[b2t_i, \{i, 10\}]\right) \text{ /. } T\_ \to T_1\right) \sim B_{Range[10]} \sim$ Product[$t2b_i$, {i, 10}]

Out[∘]= $$\mathbb{E}\Big[a_5 b_1 + a_7 b_3 + a_2 b_6 - a_8 b_8 - a_9 b_9 - a_{10} b_{10},\; \frac{1}{B_1}\left(B_1 x_5 y_1 + B_1 x_7 y_3 + B_1 x_2 y_6 - x_8 y_8 - x_9 y_9 - x_{10} y_{10}\right),\; B_1 + \frac{1}{4 B_1}\Big(4 a_1 B_1^2 + 4 a_4 B_1^2 - 4 a_8 B_1^2 - 4 a_9 B_1^2 - 4 a_{10} B_1^2 - B_1^2 x_5^2 y_1^2 - B_1^2 x_7^2 y_3^2 - B_1^2 x_2^2 y_6^2 + 4 a_1 B_1 x_8 y_8 - 8 a_8 B_1 x_8 y_8 - 3 x_8^2 y_8^2 + 4 a_1 B_1 x_9 y_9 - 8 a_9 B_1 x_9 y_9 - 3 x_9^2 y_9^2 + 4 a_1 B_1 x_{10} y_{10} - 8 a_{10} B_1 x_{10} y_{10} - 3 x_{10}^2 y_{10}^2\Big) \epsilon + O[\epsilon]^2\Big]$$

```
Timing[
 Do[Z = Z~B1,r~dm1,r→1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[◦]= $\left\{5.39063, \frac{B_1}{1-B_1+B_1^2}+\left(B_1\left(-B_1+2B_1^2+2B_1^4+a_1\left(-1+B_1-B_1^3+B_1^4\right)-2x_1y_1-B_1^3\left(3+2x_1y_1\right)\right)\epsilon\right)\Big/\left(1-B_1+B_1^2\right)^3+O[\epsilon]^2\right\}$

In[◦]:=
```
Timing[
 Z = R1,5 R6,2 R3,7 CC̄4 Kink̄8 Kink̄9 Kink̄10 /. B_ → B1;
 Do[Print["doing ", r]; Z = Z ~ B1,r ~ dm1,r→1 /. B_ → B1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[◦]= $\left\{5.3125, \frac{B_1}{1-B_1+B_1^2}+\frac{1}{2B_1\left(1-B_1+B_1^2\right)^3}\left(2a_1B_1^2\left(-1+B_1-B_1^3+B_1^4\right)-6x_1^2y_1^2+4B_1^7x_1^2y_1^2-2B_1^8x_1^2y_1^2+B_1^2x_1y_1\left(5-6x_1y_1\right)+3B_1x_1y_1\left(-1+2x_1y_1\right)+B_1^6\left(3+3x_1y_1-6x_1^2y_1^2\right)-B_1^5\left(4+13x_1y_1+2x_1^2y_1^2\right)+B_1^4\left(2+15x_1y_1+4x_1^2y_1^2\right)-B_1^3\left(1+15x_1y_1+6x_1^2y_1^2\right)\right)\epsilon+O[\epsilon]^2\right\}$